

9/4.8

Représentation de courbes

Dans ce chapitre, nous vous présentons trois programmes écrits en Basic qui permettent de représenter :

- des courbes monovariées d'équation $Y = f(x)$;
- des courbes polaires d'équation $r = f(\theta)$;
- des courbes paramétriques d'équations
 $X = f(t)$
 et $Y = g(t)$

La partie « utilisateur » a été simplifiée au maximum. Il suffit d'entrer l'équation ou les équations des courbes à représenter. Les facteurs d'échelle sont calculés automatiquement.

9/4.8.1

Courbes d'équation $Y = f(x)$

COMMENT UTILISER LE PROGRAMME

Saisissez le programme ci-dessous :

```

1000 ' -----
1010 ' Trace d'une courbe d'équation Y=f(x)
1020 ' avec recherche de l'échelle optimale
1030 ' -----
1040 '
1050 MODE 2
1060 PRINT"Etude d'une fonction Y=f(x)"
1070 PRINT"-----"
1080 PRINT
1090 PRINT"Tapez"
1100 PRINT" 2050 y="
1110 PRINT"suivi de la fonction a etudier."
1120 PRINT
1130 PRINT"Par exemple :"
1140 PRINT" 2050 y=sin(x)"
1150 PRINT
1160 PRINT"Tapez ensuite"
1170 PRINT" run 1220"
1180 PRINT"pour effectuer le trace"
1190 PRINT
1200 STOP
1210 '
1220 DEG
1230 '
1240 ' -----
1250 ' Saisie de l'intervalle d'étude
1260 ' -----
1270 '
1280 PRINT"Entrez l'intervalle d'étude : "
1290 INPUT" Minimum : ";mi
1300 INPUT" Maximum : ";ma
1310 '
1320 ' -----
1330 ' Calcul du min et du max en Y
1340 ' -----

```

```

1350 '
1360 u=-1E+15 'Maximum en X
1370 l=1E+15 'Minimum en Y
1380 FOR x=mi TO ma STEP (ma-mi)/100
1390   GOSUB 2000
1400   IF y<l THEN l=y:xl=x
1410   IF y>u THEN u=y:xu=x
1420 NEXT x
1430 '
1440 ' - - - - -
1450 ' Trace de la courbe
1460 ' - - - - -
1470 '
1480 CLS
1490 LOCATE 16,1
1500 PRINT"Trace de courbes monovariabiles d'equation Y=f(x)"

1510 i=0
1520 x=mi : GOSUB 2000: ct=345-(291*1/(1-u))
1530 MOVE 50,50 : DRAW 589,50 : DRAW 589,349 : DRAW 50,349 :
  DRAW 50,50
1540 MOVE 54,54 : DRAW 585,54 : DRAW 585,345 : DRAW 54,345 :
  DRAW 54,54
1550 FOR x=mi TO ma STEP (ma-mi)/100
1560   GOSUB 2000
1570   IF x=mi THEN MOVE 54,399-(ct+(Y*291)/(1-u))
1580   IF x<>mi THEN DRAW 54+5.31*I,399-(ct+(Y*291)/(1-u))

1590   i=i+1
1600 NEXT x
1610 MOVE 54,ct : DRAW 585,ct 'Axe des X
1620 '
1630 ' - - - - -
1640 ' Affichage des min et max
1650 ' - - - - -
1660 '
1670 LOCATE 10,24
1680 PRINT "Y minimum : ";l;" pour x=";xl
1690 LOCATE 10,25
1700 PRINT "Y maximum : ";u;" pour x=";xu
1710 a$=INKEY$: IF a$="" THEN 1710
1720 CLS
1730 END

2000 '
2010 ' -----
2020 ' Equation a etudier
2030 ' -----
2040 '
2050 y=SIN(x)*SIN(3*x)
2060 RETURN

```

Lancez le programme en tapant

RUN

L'écran se présente alors comme suit :

Etude d'une fonction $Y = f(x)$

Tapez

2050 y =

suivi de la fonction à étudier

Par exemple :

2050 y = sin(x)

Tapez ensuite

run 1220

pour effectuer le tracé

Break in 1200

Ready

Suivez les indications affichées sur l'écran.

Inscrivez la fonction à étudier en ligne 2050 et lancez le programme en tapant :

RUN 1220

Le programme vous demande alors d'entrer les bornes du domaine d'étude en **X** :

Entrez l'intervalle d'étude :

Minimum : ?

Maximum : ?

Remarque :

Si vous étudiez des fonctions sinusoïdales, le domaine d'étude doit être exprimé en degrés.

Lorsque les deux valeurs sont entrées, le programme prend quelques secondes pour calculer le facteur d'échelle en **Y**. Il représente ensuite la fonction en mode 2.


Pour quitter le programme, appuyer sur une touche quelconque du clavier.

Exemple :

Voici le résultat obtenu pour l'équation

Y = SIN(x) * SIN(3*x)


entre 0 et 360 degrés :



Les valeurs minimales et maximales en Y sont précisées en-dessous de la courbe. Lorsque cela est possible, une ligne horizontale visualise l'axe des abscisses.

LE PROGRAMME EN DÉTAIL

La logique du programme apparaît dans l'ordinogramme suivant :



Les lignes 1000 à 1200 affichent l'écran de présentation qui apparaît à chaque exécution du programme.

La ligne 1220 initialise les calculs d'angles en degrés. Les lignes 1240 à 1300 permettent de définir les bornes de l'intervalle d'étude.

Pour calculer le facteur d'échelle en Y , le programme parcourt la courbe sur tout l'intervalle d'étude et mémorise les valeurs minimale et maximale obtenues.

Voyons en détail quel est le processus utilisé. Les variables u et l contiennent respectivement le maxima et le minima en Y de la courbe sur l'intervalle d'étude. Ces variables sont initialisées avec des valeurs telles que toute valeur calculée à partir de l'équation de la courbe soit supérieure au maxima et inférieure au minima :

```
1360 u = - 1E + 15
```

```
1370 l = 1E + 15
```

La courbe est parcourue sur 100 points à l'aide d'une boucle FOR NEXT. Toute valeur supérieure au maxima devient le nouveau maxima, et toute valeur inférieure au minima devient le nouveau minima :

```
1380 FOR x = mi to ma STEP (ma-mi) / 100
```

```
1390     GOSUB 2000
```

```
1400     IF y < 1 THEN 1 = y : x1 = x
```

```
1410     IF y > u THEN u = y : xu = x
```

```
1420 NEXT x
```

Le programme se poursuit par le tracé de la courbe. Les lignes 1530 et 1540 délimitent l'aire de tracé par un cadre rectangulaire.

Les lignes 1550 à 1600 effectuent le tracé. Chacune des 100 valeurs est calculée par un appel au sous-programme situé en 2000 :

```
1550 FOR x = mi to ma STEP (ma-mi) / 100
```

```
1560     GOSUB 2000
```

Le premier point est positionné sur l'écran à l'aide d'une instruction PLOT (ligne 1570).

Les points suivants sont reliés à leurs prédécesseurs à l'aide d'une instruction LINE (ligne 1580).

Le tracé se termine par l'affichage de l'axe des abscisses ligne 1610 :
MOVE 54,ct : DRAW 585,xt 'Axe des X

et par l'affichage des valeurs extrémales en Y (lignes 1670 à 1720).

